

Fitxers per lots (batch)

Fins ara la major part del treball en MS-DOS era interactiu, és a dir, l'usuari introduïa les ordres, i n'observava el resultat de forma pràcticament immediata. Els processos per lots, o processos batch ens proporcionen una forma diferent de treballar.

Un fitxer de processos per lots o fitxer batch, és un arxiu de text en ascii que contenen ordres o fitxers executables, que s'executaran automàticament sense la intervenció de l'usuari.

Aquests fitxers han de tenir extensió bat i só executables directament des de l'indicador del sistema. A més de les ordres pròpies del sistema, també es poden utilitzar una sèrie d'ordres específiques per als processos batch, que en permeten potenciar el funcionament.

Per tal de crear un fitxer bat, seguirem els passos següents:

- Un fitxer bat es pot crear utilitzant qualsevol editor o processador que no utilitzi caràcters de control, per exemple l'EDIT que du incorporat l'MS-DOS. Un cop escrit s'ha de desar donant-li un nom i obligatòriament amb l'extensió bat.
- Un cop s'ha executat el fitxer bat, per executar només cal escriure el nom del fitxer sense l'extensió en el símbol del sistema, i prémer <RETURN>, o <ENTER>
- En qualsevol moment podem aturar l'execució d'un fitxer bat, per fer-ho cal prémer al mateix temps les tecles <ctrl> i <c>. Apareix un missatge demanant la confirmació del tipus "vols acabar el procés per lots (S/N)", i triem la opció que ens interressi.

Ordres específiques dels fitxers batch

El sistema operatiu inclou una sèrie de característiques que permeten augmentar les prestacions i la potència d'aquest tipus de fitxers. A més a més de les ordres que ja eren presents en el fitxer autoexec.bat (REM, ECHO, @), en tenim d'altres, però abans de comentar-les veurem alguns exemples de la utilització de les ordres que ja eren presents en l'autoexec.bat.

Exemple d'utilització de l'ordre REM

```
REM procés per lots
REM esborra la pantalla
REM data i hora
CLS
DATE
TIME
```

Exemple d'utilització de l'ordre ECHO

```
ECHO OFF
REM procés per lots
REM 3-2-2003
```

```
REM Esborra la pantalla
REM demana la data i la hora
ECHO esborra la pantalla...
CLS
ECHO ara podeu modificar l'hora
TIME
ECHO ara podeu modificar la data
DATE
```

Exemple d'utilització de l'ordre @

```
@ REM procés per lots
@ REM esborra la pantalla
@ REM demana la data
@ CLS
@ DATE
```

Exemple utilització de les ordres ECHO i @

```
@ECHO OFF
REM procés per lots
REM esborra la pantalla
REM demana la data
CLS
DATE
```

Les ordres que poden aparèixer en els fitxers bat en general, i que en principi encara no han aparegut al fitxer de configuració autoexec.bat són les següents:

PAUSE

Format: PAUSE

Significat: Atura momentàniament l'execució del fitxer bat i treu per pantalla el text : "presione cualquier tecla para continuar". És possible de fer que s'aturi momentàniament l'execució del programa sense que sigui necessari mostrar per pantalla el missatge dit abans. Això ho podem fer de la forma següent:

```
ECHO Premeu qualsevol tecla per continuar...
PAUSE >NUL
```

Exemples d'utilització de l'ordre PAUSE:

Exemple 1.

```
@ECHO OFF
REM copia de fitxers amb l'extensió doc
REM des del directori ALFA de la unitat C
REM fins a l'arrel de la unitat A
CLS
ECHO.
ECHO introdueix un disquet buit en la unitat A:
ECHO.
PAUSE
```

```
COPY C:\ALFA\*.DOC A:\
CLS
```

Exemple 2.

```
@ECHO OFF
CLS
ECHO Premeu una tecla per continuar
PAUSE >NUL
CLS
```

GOTO

Format: GOTO :etiqueta

Significat: Permet canviar el sentit d'execució de les ordres d'un fitxer per lots. El canvi de sentit és un salt incondicional a una línia marcada amb una etiqueta marcada en l'ordre. L'etiqueta es col·loca al principi de la línia que ens interessa i comença per dos punts (:)

Exemple utilització ordre GOTO

```
@ECHO OFF
REM Exemple de goto
CLS
GOTO :FI
ECHO no executat
:FI
ECHO executada!
ECHO premeu una tecla
PAUSE >NUL
CLS
```

IF

L'ordre IF permet la comprovació de condicions durant l'execució d'un procés per lots. Si la condició és certa, executarà l'ordre que segueix a la condició, en cas contrari s'ignora l'ordre i es passa a la línia següent.

Formats que pot tenir l'ordre IF:

a) IF [NOT] ERRORLEVEL nombre ORDRE

Significat: La condició és donada per l'ERRORLEVEL nombre; indica que una condició només serà veritat si l'ordre executada anteriorment ha retornat un codi de sortida igual o més gran que l'indicat en la condició. Sempre que s'executa una ordre, l'MS-DOS retorna un valor que depèn de la instrucció executada i del resultat d'aquesta. Es retorna el valor 0 quan l'ordre s'ha executat correctament, i un valor diferent en cas contrari.

b) IF [NOT] CADENA1 == CADENA2 ORDRE

Significat: Fa una comparació de cadenes de caràcters. Les cadenes no cal escriure-les entre cometes dobles.

c) IF [NOT] EXIST trajectòria_fitxer ORDRE

Significat: Comprova l'existència d'un fitxer.

[NOT] : equival a negar una cosa. El seu ús és opcional.

Exemple:

```
@ECHO OFF
REM Formata un disquet
CLS
ECHO introdueix un disquet en la unitat A:
ECHO prem una tecla per continuar
PAUSE > NUL
ECHO.
FORMAT A:
IF NOT ERRORLEVEL 0 GOTO : ERROR
ECHO Disquet formatat
GOTO : FI
:ERROR
ECHO No puc formatar!
:FI
ECHO Prem una tecla per continuar
CLS
```

CHOICE

Format: CHOICE [/C:Tecles] [/N] [/S] [T:c,nn] [Text]

Significat : Permet a l'usuari fer una selecció respecte d'un conjunt d'opcions que ofereix el procés per lots.

Opcions:

- /C Tecles : indica les tecles que podem utilitzar; visualment apareixen separades per comes i entre claudàtors, seguides d'un signe d'interrogació. Si no s'utilitza aquesta opció apareix [S/N] com a valor predeterminat.
- /N : Evita que choice presenti el conjunt d'opcions que ofereix per defecte.
- /S: diferencia entre minúscules i majúscules
- /T : c,nn : Serveix per a indicar que esperi la resposta de l'usuari durant un temps màxim de nn segons, i si passat aquest temps l'usuari no ha triat cap de les opcions, s'agafa la opció especificada en c.
- /Text: és el text que es presenta a l'usuari abans de les possibles opcions.
- L'ordre retorna un valor que pot ser controlat mitjançant condicions de tipus ERRORLEVEL. Per al primer caràcter retorna el valor 1, per al segon

retorna el valor 2, i així successivament. Si es produeix qualsevol error en la seva execució retorna el valor 255.

Exercici 1.

Descriu què fa aquest fitxer batch línia per línia, i després digues quin és el seu objectiu de forma global.

```
@ECHO OFF
REM aplicació de l'ordre CHOICE
REM nom del fitxer : menu.bat
REM data de creació : 3-2-2003
REM gestiona un menú d'opcions
:REPETIR
CLS
REM presentació opcions
ECHO.
ECHO MENU OPCIONS
ECHO =====
ECHO.
ECHO.
ECHO 1. OPCIÓ PRIMERA
ECHO 2. OPCIÓ SEGONA
ECHO 3. OPCIÓ TERCERA
ECHO 4.
ECHO 5.
ECHO 0. SORTIR AL SISTEMA
ECHO.
REM Demanem l'opció
CHOICE /C:123450 /TEXT:Selecciona (1,2,3,4,5,0):
REM control opció escollida
IF ERRORLEVEL 255 GOTO :ERROR
IF ERRORLEVEL 6 GOTO :FI
IF ERRORLEVEL 5 GOTO :NACTIVA
IF ERRORLEVEL 4 GOTO :NACTIVA
IF ERRORLEVEL 3 GOTO :TRES
IF ERRORLEVEL 2 GOTO :DOS
IF ERRORLEVEL 1 GOTO :U
REM Control
:ERROR
ECHO Error en el procés
ECHO Prem una tecla per continuar
PAUSE>NUL
GOTO :FI
:NACTIVA
ECHO Opció no disponible, premeu una tecla per continuar
PAUSE>NUL
GOTO :FI
:TRES
ECHO Opció triada = 3
ECHO Premeu una tecla per continuar
PAUSE>NUL
GOTO :FI
:DOS
ECHO Opció triada = 2
ECHO Premeu una tecla per continuar
PAUSE>NUL
GOTO :FI
:U
ECHO Opció triada = 1
```

```
ECHO Premeu una tecla per continuar
PAUSE>NUL
GOTO :FI
:FI
ECHO Procés acabat
ECHO Premeu una tecla per acabar
PAUSE>NUL
CLS
```

FOR

Format: FOR %%variable IN (patró) DO ordre

On: %%variable és un símbol que anirà agafant els valors des del patró, està format per dos %% i una lletra

Patró: indica la plantilla o plantilles que s'aniran assignant a la variable.

Significat: Repeteix una ordre el nombre de vegades que indica el nombre total de valors que hi ha en el patró.

Exemple

```
@ECHO OFF
REM Exemple de l'ordre FOR
REM
CLS
FOR %%A IN (1 2 5 7) DO ECHO Informàtica
ECHO Fi del procés, premeu una tecla per acabar.
PAUSE>NUL
CLS
```

Aquest fitxer batch escriurà quatre vegades la paraula informàtica per pantalla.

CALL

Format: CALL nom_fitxer_bat (paràmetre1, paràmetre2,...)

Significat: Permet cridar un altre procés per lots sense para l'execució del primer. Amb la crida també podem passar paràmetres.

Exemple:

```
1.BAT

@ECHO OFF
REM .....
REM .....
CLS
ECHO Això és el bat: 1
CALL 2.BAT
ECHO Torno a ser en el primer bat
ECHO Prem una tecla per acabar
PAUSE>NUL
```

```
2.BAT
```

```
@ECHO OFF
REM .....
REM
CLS
ECHO Això és el bat: 2
ECHO Prem una tecla per continuar
PAUSE>NUL
CALL 3.BAT
ECHO Estem en el bat 2
ECHO Prem una tecla per continuar
PAUSE>NUL
CLS
```

```
3.BAT
```

```
@ECHO OFF
REM .....
REM .....
CLS
ECHO Això és el bat: 3
ECHO Prem una tecla per continuar
PAUSE>NUL
CLS
```

Pas de paràmetres a fitxers bat.

Tots els exemples de fitxers batch que hem descrit fins ara, estaven formats per ordres fixes, és a dir, el contingut del fitxer bat no canviava, però això no sempre ha de ser d'aquesta manera; existeix la possibilitat de passar arguments o paràmetres al fitxer des de la línia de comandes de l'MS-DOS.

Format: nom_del_fitxer_bat par1 par2, ...

par1, par2,... representen els arguments o paràmetres que passem al fitxer bat. Un paràmetre pot ser de diferents tipus i tenir diferents valors. Pot ser una cadena de caràcters, un número...

Significat: Passa informació al fitxer bat des de la línia de comandes, o des d'un altre fitxer bat que crida al primer.

Els paràmetres dins dels fitxers bats es representen mitjançant una variable que té el format %nombre.

Format: %nombre %1 %2 %3 %n

Significat: són les variables que s'utilitzen dins d'un fitxer bat per a gestionar els paràmetres passats des de la línia de comandes o des d'un altre fitxer per lots. El nombre màxim de paràmetres que podem passar a un fitxer bat és de 9.

Exercici 2: Interpreta el significat de cada línia, i digues quin seria el resultat de l'execució del següent fitxer bat si li passéssim els paràmetres 20, 50, TAULA, és a dir, si escrivíssim passa 20 50 TAULA, des de la línia de comandes.

Fitxer : passa.bat

```
@ECHO OFF
REM Pas de paràmetres
CLS
ECHO El primer paràmetre és %1
ECHO El segon paràmetre és %2
ECHO El tercer paràmetre és %3
ECHO prem una tecla per acabar...
PAUSE>NUL
CLS
```

En principi podem utilitzar com a màxim 9 paràmetres amb un fitxer bat, el problema pot sorgir quan haguem d'utilitzar-ne més de 9. Per tal de resoldre aquesta situació, podem utilitzar l'ordre SHIFT.

Format: SHIFT

Significat: Desplaça una posició a l'esquerra cada paràmetre.

Exemple:

Fitxer: impre.bat

```
@ECHO OFF
CLS
:REPE
IF "%1" == "" GOTO :FI
TYPE %1 >PRN
SHIFT
GOTO :REPE
:FI
CLS
```

Si l'executem de la següent forma:

```
C:\>IMPRE a b c d e f g h i j k l m n o p q r s
```

On a b c... són nom de fitxers de text en codi ascii, el resultat de l'execució del programa serà la impressió de tots aquests fitxers a la impressora.